// File: contracts/strategies/goldfinch/StrategyGoldfinchStaking.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/math/SafeMath.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/token/ERC20/SafeERC20.sol";
import "@openzeppelin/contracts/utils/ReentrancyGuard.sol";

import "../../interfaces/common/IUniswapRouterETH.sol";
import "../../interfaces/curve/ICurveSwap.sol";
import "../../interfaces/goldfinch/IStakingGoldfinch.sol";
import "../../libraries/SafeCurveSwap.sol";
import "../../libraries/SafeUniswapRouter.sol";
import "../../managers/FeeManager.sol";
import "../../managers/SlippageManager.sol";
import "../../managers/StratManager.sol";
import "../../utils/AddressUtils.sol";

contract StrategyGoldfinchStaking is StratManager, FeeManager, SlippageManager, ReentrancyGuard {

    using SafeERC20 for IERC20;
    using SafeMath for uint256;
    using SafeCurveSwap for ICurveSwap;
    using SafeUniswapRouter for IUniswapRouterETH;

    // Tokens used
    address public want;
    address public native;
    address public output;
    address public preferredUnderlyingToken;

    // NFT token id obtained with first deposit
    uint256 tokenId;

    // Own contracts
    address public stakingReward;

    IStakingGoldfinch.StakedPositionType public immutable stakedPositionType;

    address public pool;
    uint public poolSize;
    uint public immutable preferredUnderlyingTokenIndex;

    bool public harvestOnDeposit;
    uint256 public lastHarvest;

    // Routes
    address[] public outputToNativeRoute;
    address[] public nativeToPreferredUnderlyingTokenRoute;

    mapping(address => bool) public underlyingTokenAndFlag;
    mapping(address => uint256) public underlyingTokenAndIndex;

    event Harvest(address indexed harvester, uint256 wantHarvested, uint256 tvl);
    event Deposit(uint256 tvl);
    event Withdraw(uint256 tvl);
    event ChargeFees(uint256 harvestCallFeeAmount, uint256 strategistFeeAmount, uint256 performanceFeeAmount);
    event OwnerOperation(address indexed invoker, string method);

    constructor (
        address _want,
        address _stakingReward,
        IStakingGoldfinch.StakedPositionType _stakedPositionType,
        address _pool,
        uint _poolSize,
        address[] memory _outputToNativeRoute,
        address[] memory _nativeToPreferredUnderlyingTokenRoute,
        address _vault,
        address _unirouter,
        address _strategist,
        address _companyFeeRecipient,
        address _preferredUnderlyingToken
    ) StratManager(_strategist, _unirouter, _vault, _companyFeeRecipient) public {
        want = AddressUtils.validateOneAndReturn(_want);
        stakingReward = AddressUtils.validateOneAndReturn(_stakingReward);
        stakedPositionType = _stakedPositionType;

        pool = AddressUtils.validateOneAndReturn(_pool);
        poolSize = _poolSize;

        AddressUtils.validateManyAndReturn(_outputToNativeRoute);
        AddressUtils.validateManyAndReturn(_nativeToPreferredUnderlyingTokenRoute);

        output = _outputToNativeRoute[0];
        native = _outputToNativeRoute[_outputToNativeRoute.length - 1];
        outputToNativeRoute = _outputToNativeRoute;

        require(_nativeToPreferredUnderlyingTokenRoute[0] == native, '_nativeToPreferredUnderlyingTokenRoute[0] != native');
        nativeToPreferredUnderlyingTokenRoute = _nativeToPreferredUnderlyingTokenRoute;

        preferredUnderlyingToken = AddressUtils.validateOneAndReturn(_preferredUnderlyingToken);

        for (uint256 index = 0; index < poolSize; index++) {
            address tokenAddress = ICurveSwap(pool).coins(index);
            underlyingTokenAndFlag[tokenAddress] = true;
            underlyingTokenAndIndex[tokenAddress] = index;
        }

        preferredUnderlyingTokenIndex = underlyingTokenAndIndex[preferredUnderlyingToken];

        _giveAllowances();
    }

    // puts the funds to work
    function deposit() public whenNotPaused nonReentrant {
        uint256 wantBal = balanceOfWant();

        if (wantBal == 0) {
            return;
        }

        if (tokenId == 0) {
            tokenId = IStakingGoldfinch(stakingReward).stake(wantBal, stakedPositionType);
        } else {

            uint256 stakedBalance = IStakingGoldfinch(stakingReward).stakedBalanceOf(tokenId);

            IStakingGoldfinch(stakingReward).unstake(tokenId, stakedBalance);

            IStakingGoldfinch(stakingReward).getReward(tokenId);

            uint256 updatedWantBal = balanceOfWant();

            tokenId = IStakingGoldfinch(stakingReward).stake(updatedWantBal, stakedPositionType);
        }

        emit Deposit(balanceOf());
    }

    function withdraw(uint256 _amount) external nonReentrant {
        require(msg.sender == vault, "!vault");

        if (tx.origin != owner() && !paused()) {
            uint256 withdrawalFeeAmount = _amount.mul(withdrawalFee).div(FEE_DENOMINATOR);
            _amount = _amount.sub(withdrawalFeeAmount);
        }

        uint256 wantBal = balanceOfWant();
        if (wantBal < _amount) {
            IStakingGoldfinch(stakingReward).unstake(tokenId, _amount.sub(wantBal));
        }

        IERC20(want).safeTransfer(vault, _amount);

        emit Withdraw(balanceOf());
    }

    // Compounds earnings and charges fees.
    function harvest() public whenNotPaused {
        if (tokenId == 0) {
            return;
        }

        IStakingGoldfinch(stakingReward).getReward(tokenId);

        uint256 outputBal = IERC20(output).balanceOf(address(this));

        if (outputBal == 0) {
            return;
        }

        IUniswapRouterETH(unirouter).safeSwapExactTokensForTokens(slippage, outputBal, outputToNativeRoute, address(this), now);

        _chargeFees();
        _addLiquidity();
        uint256 wantHarvested = balanceOfWant();
        deposit();

        lastHarvest = block.timestamp;
        emit Harvest(msg.sender, wantHarvested, balanceOf());
    }

    function beforeDeposit() external override {
        if (harvestOnDeposit) {
            require(msg.sender == vault, "!vault");
            harvest();
        }
    }

    function setHarvestOnDeposit(bool _harvestOnDeposit) external onlyOwner {
        harvestOnDeposit = _harvestOnDeposit;

        if (harvestOnDeposit) {
            setWithdrawalFee(0);
        } else {
            setWithdrawalFee(withdrawalFee);
        }

        emit OwnerOperation(msg.sender, "StrategyGoldfinchStaking.setHarvestOnDeposit");
    }

    // pauses deposits and withdraws all funds from third party systems.
    function panic() external onlyOwner {
        pause();
        IStakingGoldfinch(stakingReward).unstake(tokenId, balanceOfPool());

        emit OwnerOperation(msg.sender, "StrategyGoldfinchStaking.panic");
    }

    function pause() public onlyOwner {
        _pause();

        _removeAllowances();

        emit OwnerOperation(msg.sender, "StrategyGoldfinchStaking.pause");
    }

    function unpause() external onlyOwner {
        _unpause();

        _giveAllowances();

        deposit();

        emit OwnerOperation(msg.sender, "StrategyGoldfinchStaking.unpause");
    }

    // calculate the total underlying 'want' held by the strat.
    function balanceOf() public view returns (uint256) {
        return balanceOfWant().add(balanceOfPool());
    }

    // it calculates how much 'want' this contract holds.
    function balanceOfWant() public view returns (uint256) {
        return IERC20(want).balanceOf(address(this));
    }

    // it calculates how much 'want' the strategy has working in the farm.
    function balanceOfPool() public view returns (uint256) {
        return IStakingGoldfinch(stakingReward).stakedBalanceOf(tokenId);
    }

    function coins(uint256 _index) external view returns (address){
        return ICurveSwap(pool).coins(_index);
    }

    function underlyingToken(address _tokenAddress) external view returns (bool){
        return underlyingTokenAndFlag[_tokenAddress];
    }

    function underlyingTokenIndex(address _tokenAddress) external view returns (uint256){
        return underlyingTokenAndIndex[_tokenAddress];
    }

    // Uses optimisticClaimable as it accounts for claimableRewards plus earnedSinceLastCheckpoint
    function rewardsAvailable() external view returns (uint256) {
        return IStakingGoldfinch(stakingReward).optimisticClaimable(tokenId);
    }

    receive() external payable {}

    // Charge harvest call, strategy, and performance fees from profits earned.
    function _chargeFees() internal {
        uint256 nativeBal = IERC20(native).balanceOf(address(this));

        uint256 harvestCallFeeAmount;
        if (harvestCallFee > 0) {
            harvestCallFeeAmount = nativeBal.mul(harvestCallFee).div(FEE_DENOMINATOR);
            IERC20(native).safeTransfer(tx.origin, harvestCallFeeAmount);
        }

        uint256 strategistFeeAmount;
        if (strategistFee > 0 && strategist != address(0)) {
            strategistFeeAmount = nativeBal.mul(strategistFee).div(FEE_DENOMINATOR);
            IERC20(native).safeTransfer(strategist, strategistFeeAmount);
        }

        uint256 performanceFeeAmount = nativeBal.mul(performanceFee).div(FEE_DENOMINATOR);
        IERC20(native).safeTransfer(companyFeeRecipient, performanceFeeAmount);

        emit ChargeFees(harvestCallFeeAmount, strategistFeeAmount, performanceFeeAmount);
    }

    function _addLiquidity() internal {
        uint256 nativeBal = IERC20(native).balanceOf(address(this));

        IUniswapRouterETH(unirouter).safeSwapExactTokensForTokens(slippage, nativeBal, nativeToPreferredUnderlyingTokenRoute, address(this), now);

        uint256 preferredUnderlyingTokenBal = IERC20(preferredUnderlyingToken).balanceOf(address(this));

        uint256[2] memory amounts;
        amounts[preferredUnderlyingTokenIndex] = preferredUnderlyingTokenBal;
        ICurveSwap(pool).safeAddLiquidityUsingNoDepositSlippageCalculation(slippage, amounts);
    }

    function _giveAllowances() internal {
        IERC20(native).safeApprove(unirouter, type(uint).max);
        IERC20(output).safeApprove(unirouter, type(uint).max);
        IERC20(want).safeApprove(stakingReward, type(uint256).max);
        IERC20(preferredUnderlyingToken).safeApprove(pool, type(uint).max);
    }

    function _removeAllowances() internal {
        IERC20(native).safeApprove(unirouter, 0);
        IERC20(output).safeApprove(unirouter, 0);
        IERC20(want).safeApprove(stakingReward, 0);
        IERC20(preferredUnderlyingToken).safeApprove(pool, 0);
    }

}

// File: contracts/libraries/SafeCurveSwap.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/math/SafeMath.sol";

import "../interfaces/curve/ICurveSwap.sol";

library SafeCurveSwap {

    using SafeMath for uint256;

    // 2 length array of amounts

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[2] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount);
    }

    function safeAddLiquidity(
        ICurveSwap _pool,
        uint[2] memory _slippage,
        uint256[2] memory _amounts,
        uint256 _depositNativeAmount
    ) internal {

        uint256 minMintAmount = _pool.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _pool.add_liquidity{value : _depositNativeAmount}(_amounts, minMintAmount);
    }

    function safeAddLiquidityUsingNoDepositSlippageCalculation(
        ICurveSwap _pool,
        uint[2] memory _slippage,
        uint256[2] memory _amounts
    ) internal {

        uint256 minMintAmount = _pool.calc_token_amount(_amounts).mul(_slippage[0]).div(_slippage[1]);

        _pool.add_liquidity(_amounts, minMintAmount);
    }

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[2] memory _amounts,
        bool _useUnderlying
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount, _useUnderlying);
    }

    function safeAddLiquidity(
        ICurveSwap _used,
        address _pool,
        uint[2] memory _slippage,
        uint256[2] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_pool, _amounts, minMintAmount);
    }

    // 3 length array of amounts

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[3] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount);
    }

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[3] memory _amounts,
        bool _useUnderlying
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount, _useUnderlying);
    }

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        address _pool,
        uint256[3] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_pool, _amounts, minMintAmount);
    }

    // 4 length array of amounts

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[4] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount);
    }

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        address _pool,
        uint256[4] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_pool, _amounts, minMintAmount);
    }

    // 5 length array of amounts

    function safeAddLiquidity(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256[5] memory _amounts
    ) internal {

        uint256 minMintAmount = _used.calc_token_amount(_amounts, true).mul(_slippage[0]).div(_slippage[1]);

        _used.add_liquidity(_amounts, minMintAmount);
    }

    //

    function safeRemoveLiquidityOneCoin(
        ICurveSwap _used,
        uint[2] memory _slippage,
        uint256 tokenAmount,
        int128 index
    ) internal {

        uint256 minMintAmount = _used.calc_withdraw_one_coin(tokenAmount, index).mul(_slippage[0]).div(_slippage[1]);

        _used.remove_liquidity_one_coin(tokenAmount, index, minMintAmount);
    }

}

// File: contracts/libraries/SafeUniswapRouter.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/math/SafeMath.sol";

import "../interfaces/common/IUniswapRouterETH.sol";

library SafeUniswapRouter {

    using SafeMath for uint256;

    function safeSwapExactTokensForTokens(
        IUniswapRouterETH _used,
        uint[2] memory _slippage,
        uint _amountIn,
        address[] memory _path,
        address _to,
        uint deadline
    ) internal {

        uint indexAmountOut = _path.length - 1;

        uint[] memory amountsOutExpected = _used.getAmountsOut(_amountIn, _path);

        uint256 minMintAmount = amountsOutExpected[indexAmountOut].mul(_slippage[0]).div(_slippage[1]);

        uint[] memory amountsOutObtained = _used.swapExactTokensForTokens(_amountIn, minMintAmount, _path, _to, deadline);

        require(amountsOutObtained[indexAmountOut] >= minMintAmount, "amountsOutObtained[indexAmountOut]<minMintAmount");
    }
}

// File: contracts/managers/FeeManager.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "./StratManager.sol";

abstract contract FeeManager is StratManager {
    // Fees are set as a multiple of 0.01%.
    uint constant public FEE_DENOMINATOR = 10000;

    uint constant public MAX_HARVEST_CALL_FEE = 100;
    uint constant public MAX_PERFORMANCE_FEE = 450;
    uint constant public MAX_STRATEGIST_FEE = 450;
    uint constant public MAX_WITHDRAWAL_FEE = 10;

    uint public harvestCallFee = 0;
    uint public performanceFee = 200;
    uint public strategistFee = 0;
    uint public withdrawalFee = 5;

    event FeeUpdate(address indexed updater, string updateType, uint256 newValue);
    event OwnerOperation(address indexed invoker, string method);

    function setHarvestCallFee(uint256 _fee) external onlyOwner {
        require(_fee <= MAX_HARVEST_CALL_FEE, "!cap");

        harvestCallFee = _fee;

        emit FeeUpdate(msg.sender, "HarvestCallFee", _fee);
        emit OwnerOperation(msg.sender, "FeeManager.setHarvestCallFee");
    }

    function setPerformanceFee(uint256 _fee) external onlyOwner {
        require(_fee <= MAX_PERFORMANCE_FEE, "!cap");

        performanceFee = _fee;

        emit FeeUpdate(msg.sender, "PerformanceFee", _fee);
        emit OwnerOperation(msg.sender, "FeeManager.setPerformanceFee");
    }

    function setStrategistFee(uint256 _fee) external onlyOwner {
        require(_fee <= MAX_STRATEGIST_FEE, "!cap");

        strategistFee = _fee;

        emit FeeUpdate(msg.sender, "StrategistFee", _fee);
        emit OwnerOperation(msg.sender, "FeeManager.setStrategistFee");
    }

    function setWithdrawalFee(uint256 _fee) public onlyOwner {
        require(_fee <= MAX_WITHDRAWAL_FEE, "!cap");

        withdrawalFee = _fee;

        emit FeeUpdate(msg.sender, "WithdrawFee", _fee);
        emit OwnerOperation(msg.sender, "FeeManager.setWithdrawalFee");
    }
}

// File: contracts/managers/SlippageManager.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/access/Ownable.sol";

abstract contract SlippageManager is Ownable {

    uint constant public DEFAULT_SLIPPAGE_NUMERATOR = 9500;
    uint constant public DEFAULT_SLIPPAGE_DENOMINATOR = 10000;

    uint[2] slippage;

    event SlippageUpdate(address indexed updater, uint slippageNumerator);
    event OwnerOperation(address indexed invoker, string method);

    constructor() public {
        slippage = [DEFAULT_SLIPPAGE_NUMERATOR, DEFAULT_SLIPPAGE_DENOMINATOR];
    }

    function setSlippage(uint _slippageNumerator) external onlyOwner {
        require(_slippageNumerator <= DEFAULT_SLIPPAGE_DENOMINATOR, "_slippageNumerator>DEFAULT_SLIPPAGE_DENOMINATOR");
        slippage[0] = _slippageNumerator;

        emit SlippageUpdate(msg.sender, _slippageNumerator);
        emit OwnerOperation(msg.sender, "SlippageManager.setSlippage");
    }
}

// File: contracts/managers/StratManager.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Pausable.sol";

import "../utils/AddressUtils.sol";

contract StratManager is Ownable, Pausable {
    /**
     * @dev Company Contracts:
     * {strategist} - Address of the strategy author/deployer where strategist fee will go.
     * {vault} - Address of the vault that controls the strategy's funds.
     * {unirouter} - Address of exchange to execute swaps.
     */
    address public strategist;
    address public unirouter;
    address public vault;
    address public companyFeeRecipient;

    event StratUpdate(address indexed updater, string updateType, address newValue);
    event OwnerOperation(address indexed invoker, string method);

    /**
     * @dev Initializes the base strategy.
     * @param _strategist address where strategist fees go.
     * @param _unirouter router to use for swaps
     * @param _vault address of parent vault.
     * @param _companyFeeRecipient address where to send Company's fees.
     */
    constructor(
        address _strategist,
        address _unirouter,
        address _vault,
        address _companyFeeRecipient
    ) public {
        // strategist can be empty
        strategist = _strategist;
        unirouter = AddressUtils.validateOneAndReturn(_unirouter);
        vault = AddressUtils.validateOneAndReturn(_vault);
        companyFeeRecipient = AddressUtils.validateOneAndReturn(_companyFeeRecipient);
    }

    /**
     * @dev Updates address where strategist fee earnings will go.
     * @param _strategist new strategist address.
     */
    function setStrategist(address _strategist) external onlyOwner {
        strategist = _strategist;

        emit StratUpdate(msg.sender, "Strategist", _strategist);
        emit OwnerOperation(msg.sender, "StratManager.setStrategist");
    }

    /**
     * @dev Updates router that will be used for swaps.
     * @param _unirouter new unirouter address.
     */
    function setUnirouter(address _unirouter) external onlyOwner {
        unirouter = _unirouter;

        emit StratUpdate(msg.sender, "Inirouter", _unirouter);
        emit OwnerOperation(msg.sender, "StratManager.setUnirouter");
    }

    /**
     * @dev Updates company fee recipient.
     * @param _companyFeeRecipient new company fee recipient address.
     */
    function setCompanyFeeRecipient(address _companyFeeRecipient) external onlyOwner {
        companyFeeRecipient = _companyFeeRecipient;

        emit StratUpdate(msg.sender, "CompanyFeeRecipient", _companyFeeRecipient);
        emit OwnerOperation(msg.sender, "StratManager.setCompanyFeeRecipient");
    }

    /**
     * @dev Function to synchronize balances before new user deposit.
     * Can be overridden in the strategy.
     */
    function beforeDeposit() external virtual {}
}

// File: contracts/utils/AddressUtils.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

library AddressUtils {

    function validateOne(address _address) internal pure {
        require(_address != address(0), 'Empty address');
    }

    function validateOneAndReturn(address _address) internal pure returns (address) {
        validateOne(_address);
        return _address;
    }

    function validateMany(address[] memory _addresses) internal pure {
        for (uint256 index = 0; index < _addresses.length; index++) {
            validateOne(_addresses[index]);
        }
    }

    function validateManyAndReturn(address[] memory _addresses) internal pure returns (address[] memory) {
        validateMany(_addresses);
        return _addresses;
    }
}

// File: contracts/interfaces/common/IUniswapRouterETH.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

interface IUniswapRouterETH {
    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);

    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);

    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);

    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);

    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);

    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: contracts/interfaces/curve/ICurveSwap.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

interface ICurveSwap {

    function calc_token_amount(uint256[2] calldata _amounts) external returns (uint256 amount);
    function calc_token_amount(uint256[3] calldata _amounts) external returns (uint256 amount);
    function calc_token_amount(uint256[4] calldata _amounts) external returns (uint256 amount);
    function calc_token_amount(uint256[5] calldata _amounts) external returns (uint256 amount);

    function calc_token_amount(uint256[2] calldata _amounts, bool _is_deposit) external returns (uint256 amount);
    function calc_token_amount(uint256[3] calldata _amounts, bool _is_deposit) external returns (uint256 amount);
    function calc_token_amount(uint256[4] calldata _amounts, bool _is_deposit) external returns (uint256 amount);
    function calc_token_amount(uint256[5] calldata _amounts, bool _is_deposit) external returns (uint256 amount);

    function calc_withdraw_one_coin(uint256 _tokenAmount, int128 _index) external view returns (uint256);

    function coins(uint256 _arg0) external view returns (address);

    // Standard Curve pool (e.g. https://arbiscan.io/address/0x7f90122bf0700f9e7e1f688fe926940e8839f353)
    function add_liquidity(uint256[2] memory _amounts, uint256 _min_mint_amount) external payable;
    function add_liquidity(uint256[3] memory _amounts, uint256 _min_mint_amount) external payable;
    function add_liquidity(uint256[4] memory _amounts, uint256 _min_mint_amount) external payable;
    function add_liquidity(uint256[5] memory _amounts, uint256 _min_mint_amount) external payable;

    // useUnderlying Curve pool (e.g. https://snowtrace.io/address/0x7f90122bf0700f9e7e1f688fe926940e8839f353)
    function add_liquidity(uint256[2] memory _amounts, uint256 _min_mint_amount, bool _use_underlying) external;
    function add_liquidity(uint256[3] memory _amounts, uint256 _min_mint_amount, bool _use_underlying) external;

    // useMetapool Curve pool (e.g. https://optimistic.etherscan.io/address/0x167e42a1c7ab4be03764a2222aac57f5f6754411)
    function add_liquidity(address _pool, uint256[2] memory _amounts, uint256 _min_mint_amount) external;
    function add_liquidity(address _pool, uint256[3] memory _amounts, uint256 _min_mint_amount) external;
    function add_liquidity(address _pool, uint256[4] memory _amounts, uint256 _min_mint_amount) external;

    function remove_liquidity_one_coin(uint256 token_amount, int128 index, uint256 _min_amount) external;
}

// File: contracts/interfaces/goldfinch/IStakingGoldfinch.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

interface IStakingGoldfinch {

    enum StakedPositionType  {
        Fidu,
        CurveLP
    }

    function addToStake(uint256 tokenId, uint256 amount) external;

    function balanceOf(address _owner) external view returns (uint256);

    function optimisticClaimable(uint256 tokenId) external view returns (uint256 rewards);

    function getReward(uint256 tokenId) external;

    function stake(uint256 _amount, StakedPositionType _positionType) external returns (uint256);

    function stakedBalanceOf(uint256 tokenId) external view returns (uint256);

    function unstake(uint256 tokenId, uint256 amount) external;

}

// File: @openzeppelin/contracts/math/SafeMath.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        uint256 c = a + b;
        if (c < a) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b > a) return (false, 0);
        return (true, a - b);
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) return (true, 0);
        uint256 c = a * b;
        if (c / a != b) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a / b);
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a % b);
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");
        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a, "SafeMath: subtraction overflow");
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        if (a == 0) return 0;
        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");
        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: division by zero");
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: modulo by zero");
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        return a - b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryDiv}.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a % b;
    }
}

// File: @openzeppelin/contracts/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor () internal {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and make it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "../../utils/Context.sol";
import "./IERC20.sol";
import "../../math/SafeMath.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin guidelines: functions revert instead
 * of returning `false` on failure. This behavior is nonetheless conventional
 * and does not conflict with the expectations of ERC20 applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20 {
    using SafeMath for uint256;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;
    uint8 private _decimals;

    /**
     * @dev Sets the values for {name} and {symbol}, initializes {decimals} with
     * a default value of 18.
     *
     * To select a different value for {decimals}, use {_setupDecimals}.
     *
     * All three of these values are immutable: they can only be set once during
     * construction.
     */
    constructor (string memory name_, string memory symbol_) public {
        _name = name_;
        _symbol = symbol_;
        _decimals = 18;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5,05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless {_setupDecimals} is
     * called.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual returns (uint8) {
        return _decimals;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address sender, address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(sender, _msgSender(), _allowances[sender][_msgSender()].sub(amount, "ERC20: transfer amount exceeds allowance"));
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].sub(subtractedValue, "ERC20: decreased allowance below zero"));
        return true;
    }

    /**
     * @dev Moves tokens `amount` from `sender` to `recipient`.
     *
     * This is internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(address sender, address recipient, uint256 amount) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        _balances[sender] = _balances[sender].sub(amount, "ERC20: transfer amount exceeds balance");
        _balances[recipient] = _balances[recipient].add(amount);
        emit Transfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply = _totalSupply.add(amount);
        _balances[account] = _balances[account].add(amount);
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        _balances[account] = _balances[account].sub(amount, "ERC20: burn amount exceeds balance");
        _totalSupply = _totalSupply.sub(amount);
        emit Transfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Sets {decimals} to a value other than the default one of 18.
     *
     * WARNING: This function should only be called from the constructor. Most
     * applications that interact with token contracts will not expect
     * {decimals} to ever change, and may work incorrectly if it does.
     */
    function _setupDecimals(uint8 decimals_) internal virtual {
        _decimals = decimals_;
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be to transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual { }
}

// File: @openzeppelin/contracts/token/ERC20/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./IERC20.sol";
import "../../math/SafeMath.sol";
import "../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "../utils/Context.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () internal {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: @openzeppelin/contracts/utils/Pausable.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor () internal {
        _paused = false;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        require(!paused(), "Pausable: paused");
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        require(paused(), "Pausable: not paused");
        _;
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.2 <0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}